
Figure 1.  Multiple-feedback band-pass active filter

A) Introduction

Op-amps look nearly ideal in many audio applications because the gain of the amplifier
in the audio band is very large. Active filters can realize low-pass, high-pass, band-pass
and band-stop transfer functions. Higher order active filters are usually made by
cascading second order building blocks. The denominator of a second order filter is of
the form s2 + (To /Qo )s + To

2. In many audio applications, low-pass and high-pass filters
have a Qo that is usually less than one but in band-pass and band-stop filters it may be
much larger than one. All active filters have a maximum value for their To Qo product
due to the op-amps gain-bandwidth-product (GBP). If the filter operates much below
this product then the filter will look nearly ideal. The GBP is the product of the op-
amp’s low frequency open-loop gain times its !3 dB bandwidth. This application note
shows how Symbolic SPICETM can be used to evaluate and design an RC active band-
pass filter so as to reduce the effects of GBP.

The schematic shown in Fig. 1 is that of a Multiple-Feedback (MFB) Band-Pass Filter
[1-2]. One application for this type of filter is in the design of a Graphic Equalizer [3-5].

A Symbolic SPICE input file, mfb_bp.cir, is given in Table 1. This file uses Symbolic
SPICE’s definition [6] of an ideal op-amp which begins with the letters XOA.

Table 1.  Symbolic SPICE input file  mfb_bp.cir

MFB BAND-PASS FILTER
V1 1 0 AC 1 
R1 1 3
R2 3 0
R5 4 2
C3 3 4
C4 3 2
XOA 0 4 2
.END 

B) Running Symbolic SPICE

Running [6] the input file shown in Table 1, the following are the prompts. The user
responses are shown in bold:

    Symbolic SPICE - Circuit Analyzer and Approximator 
    Demo Version 3.1
    (C) Copyright 2010 by Willow Electronics, Inc.

INPUT FILE NAME [.cir] : mfb_bp

OUTPUT FILENAME [mfb_bp.det] : (hit enter)

Determinant string sorted according to orders of some variable? (y/n) : n

Numerical evaluation of the results? (y/n) : n

Discard terms if their magnitude falls below a threshold? (y/n) : n

Check and solve for second order filter functions? (y/n) : y

FILTER FUNCTION FILE NAME [mfb_bp.fun] : (hit enter)

Solve for a variable or expression? (y/n) : y

Available Unknowns:
V1 V2 V3

*Ignore nodes 5 and higher if present. They are used for internal numbering.

Valid Operators: +, -, *, /, ( ), { }, [ ]

Equation: v2

Solve for another variable or expression? (y/n) : n

C) Symbolic SPICE Determinant Output 

The output file mfb_bp.det listed in Table 2 is the matrix written by Symbolic SPICE
and the transfer function requested.

Table 2.  Symbolic SPICE Output File mfb_bp.det

MFB BAND-PASS FILTER

[0  ] [-G1                   -sC4                  sC4+sC3+G2+G1         ][V1 ]
[0  ]=[0                     -G5                   -sC3                  ][V2 ]
[1  ] [1                     0                     0                     ][V3 ]

*Ignore nodes 5 and higher if present. They are used for internal numbering.

Numerator of: v2

TERMS SORTED ACCORDING TO POWERS OF s

s**1 terms:

 - sC3*G1

************************************************

Denominator of: v2

TERMS SORTED ACCORDING TO POWERS OF s

s**2 terms:

 + sC4*sC3

s**1 terms:

 + sC4*G5 + sC3*G5

s**0 terms:

 + G5*G2 + G5*G1

************************************************

D) Symbolic SPICE Filter Function Output 

Symbolic SPICE will examine any transfer function to determine if its denominator is
second order in the Laplace operator s. If this is the case then it finds the parameters
associated with the different filter functions. Symbolic SPICE finds formulas for Qo , To
and Hi [6].

The output file mfb_bp.fun listed in Table 3 contains the filter parameters solved for
by Symbolic SPICE.

Table 3.  Symbolic SPICE Output File mfb_bp.fun

MFB BAND-PASS FILTER

SECOND ORDER FILTER PARAMETERS:

Qo is:

SQRT{( + C4*C3)*( + G2 + G1)}
----------------------------------
( + C4 + C3)*SQRT{ + G5}

Wo**2 is: 

( + G5)*( + G2 + G1)
------------------------
( + C4*C3)

****************************************

There exists a BAND PASS filter for : v2

BAND PASS GAIN (Hbp) is:

( - C3*G1)
--------------
( + G5)*( + C4 + C3)

****************************************

Symbolic SPICE’s format is a collection of strings of symbols. This is usually not how
most people view equations, so you may need to do some minor editing. Thus we have:

Another useful formula is that of the !3 dB bandwidth (radians/sec)

E) Design Procedure

Given a multiple-feedback band-pass filter, we can measure the component values or
use the nominal values and predict the filter parameters of Qo , To and Hbp with the
equations above.

Reversing this process would allow the selection of components given the filter
parameters. Since there are three filter parameters (Qo , To and Hbp ) and five
components ( R1 , R2 , C3 , C4 and R5 ) then we can pick two and solve for the rest. Note
that this is one way out of many to pick the components.

Eqns.1-4 [1] have been used to develop a design procedure:

1) Pick C3 = C4 to be a standard value

2) From Eqn. 4, R5 = 2 Qo /(To C3 )

3) From Eqn. 3, R1 = Qo /( Hbp To C3 )

4) From Eqn. 1, R2 = Qo /[(2 Qo
2  ! Hbp ) To C3 ]

Note that the last step implies that 2 Qo
2 > Hbp . If this is not true then develop a new

design procedure with C3 =/  C4.

F) Design Example

The merits of constant Qo filters for various Graphic Equalizers are discussed in [7]. In
[8], a 1-octave equalizer is designed using the design procedure in Section E with Qo =
2 and Hbp = 4. To expand this to an ultimate a-octave equalizer with 32 slide pots, a
constant Qo of 4.318 is recommended in [7]. If we use the suggested value of Hbp = 4 then
we will have a maximum gain of 12 dB.

Although we could pick the entire 32 band pass filters using these specifications from
fo = 16 Hz to 20k Hz in a-octave steps, let’s leave that for another application note and
just look at one filter to show a potential problem due to the gain-bandwidth-product
of the op-amp.

For Qo = 4.318, fo = 20k Hz and Hbp = 4 then

1) Pick C3 = C4 = 390p F

2) R5 = 2 Qo /(To C3 ) = 176.21k S   (Pick 180k S)

3) R1 = Qo /( Hbp To C3 ) = 22.027k S  (Pick 22k S)

4) R2 = Qo /[(2 Qo
2  ! Hbp ) To C3 ] = 2.6466k S (Pick 2.7k S)

Using Eqns. 1-3, we can re-calculate the filter parameters with the nearest standard
components or we could just re-run Symbolic SPICE. The revised input file
mfb_bp_std.cir is listed in Table 4.

Table 4.  Symbolic SPICE input file mfb_bp_std.cir

MFB BAND-PASS FILTER
V1 1 0 AC 1 
R1 1 3 22K
R2 3 0 2.7K
R5 4 2 180K
C3 3 4 390P
C4 3 2 390P
XOA 0 4 2
.END 

The output file mfb_bp_std.fun listed in Table 5 contains the filter parameters solved
for by Symbolic SPICE.

Figure 2.  Non-ideal op-amp

Table 5.  Symbolic SPICE Output File mfb_bp.fun

MFB BAND-PASS FILTER

SECOND ORDER FILTER PARAMETERS:

Qo is:

SQRT{( + C4*C3)*( + G2 + G1)}
----------------------------------
( + C4 + C3)*SQRT{ + G5}

 = 4.32575

Wo**2 is: 

( + G5)*( + G2 + G1)
------------------------
( + C4*C3)

fo = 19614.4Hz

****************************************

There exists a BAND PASS filter for : v2

BAND PASS GAIN (Hbp) is:

( - C3*G1)
--------------
( + G5)*( + C4 + C3)

 = -4.0909

***************************************

F) Symbolic SPICE’s Non-ideal Op-Amp 

Symbolic SPICE has a model for a non-ideal op-amp [3]. This is shown in Fig. 2 where
the output is Vin times (2B C gain-bandwidth-product / s).

The element definition for this model is : 

XNOA[name] <(+)input> <(-)input> <output> {ignore rest of line}

This is the format of a subcircuit model where for a numeric program the subcircuit
model begins with .SUBCKT and ends with .ENDS.

Symbolic SPICE ignores these subcircuit lines of code so that it can use its own
definition of gain-bandwidth-product. 

When this element is used in the input file, a new set of menu options appears. This is
because this element invokes an approximation algorithm [9] which is only valid for
second order active filters. If the user wishes to create an expression for the GBP, it
must be done with an equivalent circuit but not as a subcircuit.

Table 6 lists an edited version of the file mfb_bp_std.cir where XOA was replaced
with XNOA.

Table 6.  Symbolic SPICE input file mfb_bp_noa.cir

MFB BAND-PASS FILTER
V1 1 0 AC 1 
R1 1 3 22K
R2 3 0 2.7K
R5 4 2 180K
C3 3 4 390P
C4 3 2 390P
XNOA 0 4 2
.END 

Suppose we evaluate this filter using a low cost op-amp like a uA741 which has a gain-
bandwidth-product of about 1M Hz. Running the input file shown in Table 6, the
following are the prompts. The user responses are shown in bold:

    Symbolic SPICE - Circuit Analyzer and Approximator 
    Demo Version 3.1
    (C) Copyright 2010 by Willow Electronics, Inc.

INPUT FILE NAME [.cir] : mfb_bp_noa

Numerical evaluation of the results? (y/n) : y

Discard terms if their magnitude falls below a threshold? (y/n) : n

ERROR ANALYSIS FILE NAME [mfb_bp_noa.err] : (hit enter)

What is the GBP (in Hz) of XNOA? (y/n) : 1meg

The output file mfb_bp_noa.err listed in Table 7 is the analysis done by Symbolic
SPICE on the shifting of poles in the characteristic equation. The approximation
algorithm finds the change in fo and Qo symbolically due to the elements in the circuit,
the gain-bandwidth-product and the symbolic values of  fo and Qo .

Table 7.  Symbolic SPICE Output File mfb_bp_noa.err

MFB BAND-PASS FILTER

*************
* Qo and fo *
*************

Qo is:

SQRT{( + C4*C3)*( + G2 + G1)}
----------------------------------
( + C4 + C3)*SQRT{ + G5}

 = 4.32575

(2*PI*fo)**2 is:

( + G5)*( + G2 + G1)
------------------------
( + C4*C3)

fo = 19614.4 Hz

**********************************
* DQo/Qo = (D2-D0)foQo - Dfo/fo  *
**********************************

D2-D0: where k{i} = 1/GBP{i}

The numerator is:

+0 TIMES

 + 1

The denominator is:

( + G5*C4*C3)*( + G2 + G1)

*****************************
* Dfo/fo = (D2-D1)fo/(2Qo)  *
*****************************

D2-D1 : where k{i} = 1/GBP{i}

The numerator is:

-C3*k TIMES

 + G2 + G1

The denominator is:

( + G5)*( + C4 + C3)

*****************************
*   NUMERICAL  EVALUATION   *
*****************************

Dfo/fo = -0.0848468

DQo/Qo = 0.0848468

Thus we have that:

What this means symbolically is

What these results mean numerically is that there is a decrease in the center frequency
fo of approximately 8.48468 % and an increase in the filter Qo of approximately 8.48468
%. In other words, the effect of having a 1M Hz GBP is to shift fo from 19,614.4 to an
actual value of 19,614.4 (1 ! 0.0848468) = 17,950 Hz and to shift Qo from 4.32575 to an
actual value of 4.32575 (1 + 0.0848468) = 4.6928.

The shifting in fo and Qo shifts the bandwidth which is fo /Qo . In this case the bandwidth
was shifted from 19,614.4 /4.32575 = 4,534.3 Hz to an actual value of 17,950 /4.6928 =
3,825 Hz.

Therefore our filter is covering a different band of frequencies than we designed for.

G) PSpice Evaluation 

PSpice allows for controlled sources with Laplace expressions. A one-pole model is of the
form:

where Ao is the op-amp open-loop gain at low frequencies, fo is the op-amp’s open-loop
gain !3 dB frequency and Ao fo is the op-amp’s gain-bandwidth-product.

Let’s verify the Symbolic SPICE results with PSpice using an ideal op-amp in the first
simulation and a one-pole model for the op-amp in the second. The input file is listed
in Table 8 where Ao = 200k, fo = 5 Hz and the GBP = 1M Hz. The output is graphed in
Fig. 3. 

o o o19.615k Hz   4.0909   19.615k /4.515k = 4.3434f H Q= = =

oo 18.131k Hz   4.0398   18.131k /3.927k = 4.617of H Q= = =

Table 8.  PSpice input file bandpass.cir

BANDPASS ACTIVE FILTER
Vin 1 0 AC 1
R1 1 3 22K
R2 3 0 2.7K
R5 4 2 180K
C3 3 4 390P
C4 3 2 390P
XOA 0 4 2 IDEAL
.SUBCKT IDEAL 1 2 3 
RI 1 2 1000MEG 
EA 3 0 1 2 1000MEG 
.ENDS IDEAL
.AC DEC 20000 2K 200K 
.PROBE
.END 
BANDPASS ACTIVE FILTER
Vin 1 0 AC 1
R1 1 3 22K
R2 3 0 2.7K
R5 4 2 180K
C3 3 4 390P
C4 3 2 390P
XNOA 0 4 2 uA741
.SUBCKT uA741 1 2 3   
EA 3 0 LAPLACE {V(1,2)} =
+ {(200K)*(6.283*5)/(S + (6.283*5))}  
.ENDS uA741
.AC DEC 20000 2K 200K 
.PROBE
.END 

From Fig. 3, for the ideal op-amp the peak frequency is 19.615k Hz, the peak gain is
4.0909 and the !3 dB bandwidth is 22.002k !17.486k = 4.516k Hz.  Thus

and this closely agrees with Symbolic SPICE.

From Fig. 3, for a non-ideal op-amp the peak frequency is 18.131k Hz, the peak gain is
4.0398 and the !3 dB bandwidth is 20.201k !16.274k = 3.927k Hz.  Thus

Comparing to Symbolic SPICE, this is a difference of +1% for fo and !1.6% for Qo which
is close given that the algorithm used is always an approximation.

Figure 3.  Effect of GBP on the voltage gain

≈ − = −
+

H) Redesign 

In order to  reduce the error due to the GBP of the op-amp suppose that we revisit Eqns.
5 and 6. Since both the error in fo and Qo are inversely proportional to the GBP, we
could simply pick an op-amp with a higher GBP. For example, an NE5534 op-amp has
a GBP of 10M Hz. This would lower the error by a factor of 10, that is, there would be
a decrease in the center frequency fo of approximately 0.848468 % and an increase in the
filter Qo of 0.848468 %. In other words, the effect of having a 10M Hz GBP would be to
shift fo from 19,614.4 to an actual value of 19,614.4 (1 ! 0.00848468) = 19,448 Hz and
to shift Qo from 4.32575 to an actual value of 4.32575 (1 + 0.00848468) = 4.3625. Thus
the bandwidth would shifted from 19,614.4 /4.32575 = 4,534.3 Hz to an actual value of
19,448 /4.3625 = 4,458 Hz.

Another improvement could also be found by re-examining Eqns. 5 and 6 and
substituting in Eqn. 1. After a little algebra we can show that

From Eqns. 9 and 10, we see that the errors grow as Qo increases or fo increases. When
C4 = C3, then the error magnitude is Qo fo /GBP. So for this selection of components, we
can have an error of 1% or less whenever Qo fo < GBP/100. This type of result can be
found in all active filters and it is common to see a Qo fo product specification in an
active filter data sheet [10].

A closer examination of Eqns. 9 and 10 also suggests that picking C4 = C3 is not the best
choice to reduce errors due to op-amp GBP. If we pick C4 >> C3, then the error we found
in our bandpass filter would drop by about half.

Developing a new design procedure using Eqns.1-4 and the above :

1) Pick C3  to be a standard value

2) Pick C4 = 10 C3 

3) From Eqn. 4, R5 = 11 Qo /(10 To C3 )

4) From Eqn. 3, R1 = Qo /( 10 Hbp To C3 )

5) From Eqn. 1, R2 = Qo /[(11 Qo
2  ! 10 Hbp ) To C3 ]

Note that the last step implies that 11 Qo
2 > 10 Hbp . If this is not true then develop a

new design procedure with C4 =/  10 C3.

Redesigning for again Qo = 4.318, fo = 20k Hz and Hbp = 4 then

1) Pick C3 = 390p F

2) Pick C4 = 10 C3 = 3900p F

3) R5 = 11 Qo /(10 To C3 ) = 96.917k S  (Pick 100k S)

4) R1 = Qo /( 10 Hbp To C3 ) = 2.2027k S   (Pick 2.2k S)

5) R2 = Qo /[(11 Qo
2  ! 10 Hbp ) To C3 ] = 533.67 S (Pick 510 S)

Using Eqn. 1-3, we can re-calculate the filter parameters with the nearest standard
components or we could just re-run Symbolic SPICE. The output of
new_mfb_bp_std.cir is listed in Table 9.

Table 9.  Symbolic SPICE input file new_mfb_bp_std.cir

MFB BAND-PASS FILTER
V1 1 0 AC 1 
R1 1 3 2.2K
R2 3 0 510
R5 4 2 100K
C3 3 4 390P
C4 3 2 3900P
XOA 0 4 2
.END 

The output file mfb_bp_std.fun listed in Table 10 contains the filter parameters
solved for by Symbolic SPICE.

Table 10.  Symbolic SPICE Output File new_mfb_bp.fun

MFB BAND-PASS FILTER

SECOND ORDER FILTER PARAMETERS:

Qo is:

SQRT{( + C4*C3)*( + G2 + G1)}
----------------------------------
( + C4 + C3)*SQRT{ + G5}

 = 4.46781

Wo**2 is: 

( + G5)*( + G2 + G1)
------------------------
( + C4*C3)

fo =  20056Hz

****************************************

There exists a BAND PASS filter for : v2

BAND PASS GAIN (Hbp) is:

( - C3*G1)
--------------
( + G5)*( + C4 + C3)

 = -4.13223

****************************************

Table 11 lists an edited version of the file new_mfb_bp_std.cir where XOA was
replaced with XNOA.

Table 11.  Symbolic SPICE input file new_mfb_bp_noa.cir

MFB BAND-PASS FILTER
V1 1 0 AC 1 
R1 1 3 2.2K
R2 3 0 510
R5 4 2 100K
C3 3 4 390P
C4 3 2 3900P
XNOA 0 4 2
.END 

Suppose we evaluate this filter using the NE5534 which has a gain-bandwidth-product
of about 10M Hz. Running the input file shown in Table 11, the following are the
prompts. The user responses are shown in bold:

    Symbolic SPICE - Circuit Analyzer and Approximator 
    Demo Version 3.1
    (C) Copyright 2010 by Willow Electronics, Inc.

INPUT FILE NAME [.cir] : new_mfb_bp_noa

Numerical evaluation of the results? (y/n) : y

Discard terms if their magnitude falls below a threshold? (y/n) : n

ERROR ANALYSIS FILE NAME [new_mfb_bp_noa.err] : (hit enter)

What is the GBP (in Hz) of XNOA? (y/n) : 10meg

The output file new_mfb_bp_noa.err listed in Table 12 is the analysis done by
Symbolic SPICE on the shifting of poles in the characteristic equation.

Table 12.  Symbolic SPICE Output File mfb_bp_noa.err

MFB BAND-PASS FILTER

*************
* Qo and fo *
*************

Qo is:

SQRT{( + C4*C3)*( + G2 + G1)}
----------------------------------
( + C4 + C3)*SQRT{ + G5}

 = 4.46781

(2*PI*fo)**2 is:

( + G5)*( + G2 + G1)
------------------------
( + C4*C3)

fo =  20056 Hz

**********************************
* DQo/Qo = (D2-D0)foQo - Dfo/fo  *
**********************************

D2-D0: where k{i} = 1/GBP{i}

The numerator is:

+0 TIMES

 + 1

The denominator is:

( + G5*C4*C3)*( + G2 + G1)

*****************************
* Dfo/fo = (D2-D1)fo/(2Qo)  *
*****************************

D2-D1 : where k{i} = 1/GBP{i}

The numerator is:

-C3*k TIMES

 + G2 + G1

The denominator is:

( + G5)*( + C4 + C3)

*****************************
*   NUMERICAL  EVALUATION   *
*****************************

Dfo/fo = -0.00492835

DQo/Qo = 0.00492835

What these results mean numerically is that there is a decrease in the center frequency
fo of approximately 0.492835 % and an increase in the filter Qo of approximately
0.492835 %. In other words, the effect of having a 10 M Hz GBP is to shift fo from
20,056 to an actual value of 20,056 (1 ! 0.00492835) = 19,957 Hz and to shift Qo from
4.46781 to an actual value of 4.46781 (1 + 0.00492835) = 4.4898.

The shifting in fo and Qo shifts the bandwidth which is fo /Qo . The bandwidth is shifted
from 20,056 /4.46781 = 4,489 Hz to an actual value of 19,957 /4.4898  = 4,445 Hz.

When C4 = 10 C3, then the error magnitude is 11Qo fo /(20 GBP). So for this selection of
components, we can have an error of 1% or less whenever Qo fo < GBP/55. This is an
improvement of nearly two over the usual selection of C4 = C3 .

Let’s verify the Symbolic SPICE results with PSpice using an ideal op-amp in the first
simulation and a one-pole model for the op-amp in the second. The input file is listed
in Table 13 where Ao = 100k, fo = 100 Hz and the GBP = 10M Hz. The output is graphed
in Fig. 4.

o o o20.055k Hz   4.1322   20.055k /4.4874k = 4.4692f H Q= = =

oo 19.959k Hz   4.1309   19.959k /4.3647k = 4.573of H Q= = =

Table 13.  PSpice input file new_bandpass.cir

BANDPASS ACTIVE FILTER
Vin 1 0 AC 1
R1 1 3 2.2K
R2 3 0 510
R5 4 2 100K
C3 3 4 390P
C4 3 2 3900P
XOA 0 4 2 IDEAL
.SUBCKT IDEAL 1 2 3 
RI 1 2 1000MEG 
EA 3 0 1 2 1000MEG 
.ENDS IDEAL
.AC DEC 20000 2K 200K 
.PROBE
.END 
BANDPASS ACTIVE FILTER
Vin 1 0 AC 1
R1 1 3 2.2K
R2 3 0 510
R5 4 2 100K
C3 3 4 390P
C4 3 2 3900P
XNOA 0 4 2 NE5534
.SUBCKT NE5534 1 2 3   
EA 3 0 LAPLACE {V(1,2)} =
+ {(100K)*(6.283*100)/(S + (6.283*100))}  
.ENDS NE5534
.AC DEC 20000 2K 200K 
.PROBE
.END  

From Fig. 4, for the ideal op-amp the peak frequency is 20.055k Hz, the peak gain is
4.1322 and the !3 dB bandwidth is 22.425k !17.937k = 4.4874k Hz.  Thus

and this is virtually identical to Symbolic SPICE.

From Fig. 4, for a non-ideal op-amp the peak frequency is 19.959k Hz, the peak gain
is 4.1309 and the !3 dB bandwidth is 22.302k !17.937k = 4.3647k Hz.  Thus

Comparing to Symbolic SPICE, this is a difference of !0.01% for fo and 1.85% for Qo .

It is worth noting that the Symbolic SPICE error approximations are most accurate
when the errors Symbolic SPICE calculates are in the 1% range or less. Although
accuracy begins to depart around 10%, the terms dominating the error are still the
same. This is helpful in making design decisions.

Figure 4.  Redesigned MFB band-pass filter

I) Conclusion 

Symbolic SPICE can be used to analyze ideal op-amp circuits to find transfer functions.
It can recognize second order transfer functions in the Laplace operator s. Symbolic
SPICE can determine the filter type such as band-pass and then solve for the filter
parameters Qo ,To and Hbp . It can also estimate the effects of op-amp’s gain-bandwidth-
product on the filter parameters, To and Qo . Lastly this application note showed that
is possible to almost double the usable bandwidth of the MFB band-pass active filter
by proper selection of capacitor values.

